/*
* Licensed under the MIT License
* 
* Copyright (c) 2010 Kay Kreuning
* 
* Permission is hereby granted, free of charge, to any person obtaining a copy
* of this software and associated documentation files (the "Software"), to deal
* in the Software without restriction, including without limitation the rights
* to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
* copies of the Software, and to permit persons to whom the Software is
* furnished to do so, subject to the following conditions:
* 
* The above copyright notice and this permission notice shall be included in
* all copies or substantial portions of the Software.
* 
* THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
* IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
* FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
* AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
* LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
* OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
* THE SOFTWARE.
* 
* http://code.google.com/p/as3hyvesapi
*/
package nl.hyves.api.authorization
{
	import flash.events.Event;
	import flash.net.URLLoader;
	import flash.net.URLRequest;
	
	import nl.hyves.api.methods.PrivateMethods;
	import nl.hyvesstudio.oauth.Accesstoken;
	import nl.hyvesstudio.oauth.IOAuthService;
	import nl.hyvesstudio.oauth.authorization.IAuthorizer;
	
	/**
	 * This classes exchanges a logintoken for an accesstoken.
	 * 
	 * @author Kay Kreuning
	 */
	public class LogintokenAuthorizer implements IAuthorizer
	{
		private const API_ENDPOINT:String = "http://data.hyves-api.nl/";
		
		private var logintoken:String;
		private var oauthService:IOAuthService;
		
		/**
		 * Create a new LogintokenAuthorizer instance.
		 * 
		 * @param logintoken Logintoken to exchange for an accesstoken
		 */
		public function LogintokenAuthorizer(logintoken:String)
		{
			this.logintoken = logintoken;
		}
		
		/** @inheritDoc */
		public function authorize(oauthService:IOAuthService):void
		{
			this.oauthService = oauthService;
			
			var request:URLRequest = oauthService.createURLRequest(API_ENDPOINT, {
				ha_method: PrivateMethods.auth_accesstokenByLogintoken,
				logintoken: logintoken
			});
			
			var loader:URLLoader = new URLLoader(request);
			loader.addEventListener(Event.COMPLETE, onComplete, false, 0, true);
		}
		
		private function onComplete(event:Event):void
		{
			event.target.removeEventListener(Event.COMPLETE, onComplete);
			
			var response:XML = new XML(event.target.data);
			
			var accesstoken:Accesstoken = new Accesstoken(
				response.oauth_token.toString(),
				response.oauth_token_secret.toString(),
				response.userid.toString(),
				response.methods.toString().split(","),
				new Date(parseInt(response.expiredate.toString()) * 1000)
			);
			
			oauthService.accesstoken = accesstoken;
		}
	}
}